
Abstract. We use a mechanized verification system, PVS, to examine the argu-
ment from Anselm’s Proslogion Chapter III, the so-called “Modal Ontological
Argument.” We consider several published formalizations for the argument and
show they are all essentially similar. Furthermore, we show that the argument is
trivial once the modal axioms are taken into account.
This work is an illustration of computational philosophy and, in addition, shows
how these methods can help detect and rectify errors in modal reasoning.

1 Introduction
The Ontological Argument is a proof of the existence of God presented by St. Anselm
of Canterbury in his Proslogion (Anselm 1078). It has been studied and debated by
philosophers and theologians ever since.

The Ontological Argument has traditionally been identified with Chapter II of the
Proslogion, but in the middle years of the last century philosophers noted that Chapter
III presents an argument that could be interpreted as an alternative proof for God’s ex-
istence. This discovery is generally attributed to Malcolm (1960) although Hartshorne
had made similar observations earlier. (Hartshorne (1965, Part 1, Section 8) gives some
of the history, and also Smith (2014, Introduction).) The discovery was that the argu-
ment of Chapter III can be seen as an independent, self-contained proof; previously, and
for many commentators still today (Campbell 2018, 2019) (Sobel 2003, p. 82), Chapter
III is seen as a continuation of the argument of Chapter II.

We reproduce an English translation of Chapter III in Figure 1. Its argument, which
is found in its second paragraph and highlighted in italics, is couched in terms of beings
whose existence is necessary (i.e., “that which cannot be conceived not to exist”) and
has come to be known as Anselm’s Modal Argument, since qualifiers such as “neces-
sary” (with dual “possible”) indicate modes of truth. Philosophers since Aristotle have
attempted to construct specialized logics for reasoning about these but modern modal
logics were developed only in the 20th century (Cresswell and Hughes 1996).

Addendum to the published paper (8 June 2023).
The published paper provides no description of the path from Anselm’s informal

presentation of the argument in Figure 1 to the formal versions considered later.
The first step is to introduce g as an abbreviation for (the existence of) “that than

which nothing greater can be conceived” and to use 3 to mean “it is conceivable that.”
Then the sentence “Hence, if that, than which nothing greater can be conceived, can be

God cannot be conceived not to exist. God is that, than which nothing greater can be
conceived. That which can be conceived not to exist is not God.

And it assuredly exists so truly, that it cannot be conceived not to exist. For, it is possible
to conceive of a being which cannot be conceived not to exist; and this is greater
than one which can be conceived not to exist. Hence, if that, than which nothing
greater can be conceived, can be conceived not to exist, it is not that, than which
nothing greater can be conceived. But this is an irreconcilable contradiction. There
is, then, so truly a being than which nothing greater can be conceived to exist, that
it cannot even be conceived not to exist; and this being you are, O Lord, our God.

So truly, therefore, do you exist, O Lord, my God, that you can not be conceived not
to exist; and rightly. For, if a mind could conceive of a being better than you, the
creature would rise above the Creator; and this is most absurd. And, indeed, what-
ever else there is, except you alone, can be conceived not to exist. To you alone,
therefore, it belongs to exist more truly than all other beings, and hence in a higher
degree than all others. For, whatever else exists does not exist so truly, and hence in
a less degree it belongs to it to exist. Why, then, has the fool said in his heart, there
is no God (Psalms xiv. 1), since it is so evident, to a rational mind, that you do exist
in the highest degree of all? Why, except that he is dull and a fool?

Fig. 1: The Complete Chapter III of St. Anselm’s Proslogion (1078).

conceived not to exist, it is not that, than which nothing greater can be conceived” can
be read as 3¬g ⊃ ¬g, and the contrapositive of this is g ⊃ ¬3¬g (see first footnote for
notation).

At this point we need to decide how to interpret 3 formally and the choice is to treat
it as the “possibly” qualifier of Alethic modal logic. Then ¬3¬ becomes the “neces-
sary” qualifier 2 and the formal rendition becomes

g ⊃2g.

This has become known as “Anselm’s Principle” and, in modern language, its justi-
fication might be that if a greatest being exists, then its existence could not be contingent
(i.e., chance), because we can then conceive of a greater being (i.e., one whose existence
is necessary). Hence if a greatest being exists, then its existence must be necessary.

Next, it seems implicit that the existence of g is conceivable, and this is rendered as

3g.

We now have the two premises of the argument, and its completion requires rules
for reasoning about the modal qualifiers 3 and 2. When sound treatments for modal
logics started to be developed (in the last century), it became apparent that additional
axioms were needed to describe properties of the modal qualifiers, and that different
axioms were required according to the different interpretations of the qualifiers (e.g.,
knowledge vs. belief). Formalizations of the ontological argument use an Alethic logic
(one in which the qualifiers are interpreted as “possibly” and “necessary” or, to be more
faithful to Anselm “it is conceivable that” and “it is inconceivable that not”) and suitable
auxiliary axioms are those known as T, B, and 5, which are described in Section 2.
End of Addendum

In this paper, I subject several published formalized variants of Anselm’s Modal Ar-
gument to mechanized analysis in the PVS Verification system; specifically, I consider
versions by Eder and Ramharter (2015, Section 4.1), Kane (1984), Malcolm (1960) as
reported by Hartshorne (1961), Adams (1971) as reported by Matthews (2005), and
Hartshorne (1962, pp. 50–51). This work is an elementary example of Computational
Philosophy (Grim and Singer 2020), that is, application of mechanized computation
and reasoning to philosophical topics.

There are several other applications of automated reasoning to ontological argu-
ments, including first and higher-order treatments of Anselm’s traditional argument
(Oppenheimer and Zalta 2011; Rushby 2013, 2020), modal treatments of that argu-
ment (Rushby 2019), and Gödel’s modal argument (Benzmüller and Woltzenlogel-
Paleo 2014). These other applications use moderately advanced logical constructions,
such as definite descriptions (Oppenheimer and Zalta 2011; Rushby 2013), higher order
logic (Rushby 2020), and first order modal logic (Benzmüller and Woltzenlogel-Paleo
2014; Rushby 2019), whose mechanized support is fairly intricate and prone to errors
(Garbacz 2012).

In contrast, Anselm’s modal argument requires only propositional modal logic.
However, this is mechanized by embedding it in the classical logic of a verification
system, which requires some care. Nonetheless, these are among the simplest examples
of Computational Philosophy and can serve as an introduction to the field. I will show,
however, that despite its simplicity, this exercise delivers useful and novel results of
both a positive and negative kind—the latter being to expose a lamentable error in a
paper published in this very journal.

The structure of the paper is as follows. Section 2 provides a short introduction to the
embedding of propositional modal logic in PVS, which is the system that provides the
mechanization used here; the five variants of the Argument are presented and analyzed
in Section 3; Section 4 discusses the value of this exercise and some related analyses,
and Section 5 provides conclusions.

2 Propositional Modal Logic in PVS
Modal logics allow reasoning about various modes of truth: for example, what it means
for something to be necessarily true, or to know that something is true as opposed to
merely believing it.

The modal qualifier 2 and its dual 3 (defined as ¬2¬)1 are used to indicate expres-
sions that should be interpreted modally. All modal logics share the same basic structure
but they employ different sets of axioms and make other adjustments according to the
mode attributed to the qualifiers. For example, in an Alethic modal logic, where 2 is
interpreted as necessity, we will expect the formula 2P ⊃ P to hold: if something is
necessarily true, then it should be true. But we would not expect this formula to hold in
a Doxastic logic, where 2 is interpreted as belief. Instead, we might expect 2P ⊃ 3P
to hold: if I believe P is true, then I cannot also believe it to be false (reading 3P as
¬2¬P). There is a collection of formulas such as these that have standard names (the
two above are called T and D, respectively) and that are used in various combinations
to axiomatize different modal logics. Some of the common ones are listed below.

1 We use ¬ for negation, ∧ for conjunction, ∨ for disjunction and ⊃ for material implication.

K: 2(P ⊃ Q) ⊃ (2P ⊃2Q) [this formula is true in all modal logics],
T: 2P ⊃ P,
D: 2P ⊃3P,
B: P ⊃23P,
4: 2P ⊃22P,
5: 3P ⊃23P.

The semantics of modal logics are interpreted relative to a set of possible worlds,
so that 2P means true in all worlds and 3P means true in some world. To represent
different interpretations for the modalities, we impose an accessibility relation on possi-
ble worlds and refine the statements above to true in all accessible and some accessible
worlds, respectively.

There is a relationship between the standard modal axioms and properties of the ac-
cessibility relation. For example, T and D, mentioned above, correspond to accessibility
relations that are reflexive and serial, respectively. The modal ontological argument is
expressed in Alethic modal logic, and such logics have accessibility relations that are
equivalence relations. This corresponds to the combination of modal axioms T (reflex-
ive) and either 4 (transitive) plus B (symmetric), or 5 (Euclidean).2

We will undertake our examination of the modal ontological argument using the
PVS Verification System. Verification systems are tools from computer science that are
generally used for exploration and verification of software or hardware designs and al-
gorithms; they comprise a specification language, which is essentially a rich logic, and a
collection of powerful deductive engines (e.g., satisfiability solvers for combinations of
theories, model checkers, and automated and interactive theorem provers). In particular,
PVS has a specification language based on higher-order logic and its proof automation
is guided interactively. It is generally applied to analysis and verification of computa-
tional systems and has more than 4,000 citations. We do not describe PVS in detail here
but we do try to provide enough information to make this presentation self contained;
see PVS (constantly updated) for the PVS system description and Rushby (2013, 2018,
2019, 2020) for previous applications to the (traditional, Proslogion Chapter II) Onto-
logical Argument.

There is a standard translation from modal logic to classical first- or higher-order
logic (Wikipedia retrieved 2019) and we use this to provide a shallow embedding
into the dependently-typed higher-order classical logic of PVS in a way that preserves
much of its proof automation. The embedding is described in tutorial detail elsewhere
(Rushby 2017). The core of the embedding is shown in Figure 2; the basic idea is that
modal expressions are “lifted” so that they become functions on worlds. The function
val(x)(w) provides the valuation for propositional constant x in world w, and ∼, &, \/,
and => provide the “lifted” versions of negation, conjunction, disjunction, and material
implication, respectively. We then define the modalities as described above, and define
modal validity as truth in all worlds.

When we write a modal formula such as <> P => P in PVS, we really intend its
classical embedding, valid(<> P => P), and this is taken care of automatically using

2 The modal logic with no accessibility relation has the same theorems as one whose accessibil-
ity relation is an equivalence.

PVS
shallow_pml: THEORY

BEGIN

worlds: TYPE+

access: pred [[worlds, worlds]]

pmlformulas: TYPE = [worlds -> bool]

pvars: TYPE+

v, w: VAR worlds

x, y: VAR pvars

P, Q: VAR pmlformulas

val(x)(w): bool

∼(P)(w): bool = NOT P(w) ;

&(P, Q)(w): bool = P(w) AND Q(w) ;

\/(P, Q)(w): bool = P(w) OR Q(w) ;

=>(P, Q)(w): bool = P(w) IMPLIES Q(w) ;

2(P)(w): bool = FORALL v: access(w, v) IMPLIES P(v) ;

<>(P)(w): bool = EXISTS v: access(w, v) AND P(v) ;

dexpand: LEMMA <> P = ∼2∼ P ;

|=(w, P): bool = P(w)

valid(P): bool = FORALL w: w |= P

validval(x: pvars): bool = valid(val(x))

CONVERSION valid, val

END shallow_pml

Fig. 2: Shallow Embedding of Propositional Modal Logic in PVS.

the PVS CONVERSIONs (see Rushby (2017)) specified at the bottom of the theory, and
in some of the theories that appear later.

3 Reconstructions of the Modal Ontological Argument

We will examine five reconstructions of Anselm’s modal argument. As we will discover,
they are all variants of the same rather trivial deduction.

3.1 The Version by Eder and Ramharter

I will begin with the version of Eder and Ramharter (2015, Section 4.1). They attribute
this to Hartshorne (1962, pp. 49–51) but my reading of Hartshorne’s version is some-
what different and I will examine that later.3

The argument is presented in an Alethic modal logic: that is, one where the modali-
ties express necessity (2) and possibility (3), respectively.

Definition: g is an abbreviation for “there exists a being than which there is none
greater” (i.e., God).

Premise ER1: g ⊃2g (i.e., if God exists, then he exists necessarily).
Eder and Ramharter refer to ER1 as Anselm’s Principle and state that a more pre-
cise reading (i.e., a better match to Anselm’s language) is “if God exists, then his
nonexistence is inconceivable” (i.e., reading 2g as ¬3¬g).

Premise ER2: 3g (i.e., God’s existence is possible).
Conclusion ERC: g (i.e., God exists in a classical sense).

Before we proceed to examine the logic of this formulation, we note that it is ab-
stracted quite significantly from Anselm’s natural language description as given (in
English translation) in the second paragraph of Figure 1. Anselm speaks of beings that
can be “conceived” to exist, and also has a relation of “greater” among beings. Eder and
Ramharter’s paper is largely concerned with what it means for a logical formulation to
be a good reconstruction of an informal argument and they note that this formulation
fails several of their criteria. We will return to some of these points in Section 4.

Eder and Ramharter’s formulas ER1, ER2, and ERC are all rendered straightfor-
wardly into PVS as shown in Figure 3. The imported theory simple pml provides the
shallow embedding of propositional modal logic shown in the previous section, plus
some additional theories including the standard modal axioms.

3.1.1 Automated Proof. The formula ERC tmp states the basic conclusion, which we
attempt to prove using the following PVS proof commands.

PVS proof steps
(lemma "ER1") (lemma "ER2")

(grind :if-match nil)

(inst?)

(grind :polarity? t)

These install the two premises, then invoke the standard automated proof command
grind with instantiation of variables disabled (this just unwinds the embedding of
modal logic); next we perform heuristic instantiation on one of the premises, and finally
use grind again, with variable instantiation instructed to pay attention to the polarity
of variables.

3 Hartshorne uses the arrow→ to mean strict implication, so his version of (Hartshorne 1962, pp.
49–51) is that examined in Section 3.5. In the published paper, I stated “elsewhere, Hartshorne
(1965, p. 97) does state ‘Anselm’s Principle’ in the form used by Eder and Ramharter.” This
is incorrect, however, and I am grateful to Diego Murcia for pointing out that Hartshorne uses
strict implication in that version, too.

PVS
Eder_Ramharter: THEORY

BEGIN

IMPORTING simple_pml

g: pvars

P: VAR pmlformulas

ER1: AXIOM g => 2 g

ER2: AXIOM <> g

ER_triv: LEMMA symmetric?(access) IMPLIES (P => 2 P) IMPLIES (<> P => P)

CONVERSION+ validval

ERC_tmp: THEOREM g

ERC: THEOREM symmetric?(access) => g

END Eder_Ramharter

Fig. 3: Eder and Ramharter’s Version in PVS.

This delivers the following proof sequent.

PVS sequent
{-1} access(w!1, v!1)

{-2} val(g)(v!1)

|-------

{1} access(v!1, w!1)

[2] val(g)(w!1)

The interpretation of a PVS sequent is that the conjunction of formulas above the
|------- turnstile line should entail the disjunction of those below. Terms such as w!1
are Skolem constants, access is the accessibility relation on possible worlds, and val
is the valuation function for propositional constants, so val(g)(v!1) is the valuation
of g in world v!1. Here, we see that the proof would be completed if line -1 implied
line 1: that is, if the accessibility relation were symmetric.

Thus, we recognize that the conclusion to the argument should be modified to men-
tion symmetry, and this explains the final form for ERC shown in Figure 3, which is
proved by the same proof steps as above.

Alternatively, we could cite the standard modal axiom B (i.e., P ⊃ 23P), which
characterizes symmetric accessibility relations, as an additional premise. In this case,
ERC tmp is proved by the following commands. Notice that the instantiation into B is
not trivial (observe the negation).

PVS proof steps
(lemma "ER1") (lemma "ER2")

(lemma "B") (inst -1 "∼val(g)")

(grind :if-match nil)

(inst? -2)

(grind :polarity? t)

As can be inferred from the PVS sequent shown on a previous page, these proofs
operate directly on the possible world semantics of modal logic: essentially they ex-
pand the definitions given in Figure 2 and then perform quantificational reasoning over
the possible worlds. This is automated effectively and efficiently by the standard proof
mechanization of PVS and the intermediate proof states, displayed as sequents, are
fairly easy to interpret.4 However, it does not reproduce the style of semi-formal proofs
that typically accompany journal presentations. Accordingly, we now illustrate how
PVS can be used to examine those kinds of proofs.

3.1.2 Reconstructing a Manual Proof. Eder and Ramharter (2015, Section 4.2) pro-
vide a typical semi-formal manual proof that is presented below in a slightly reorganized
form.

Step 1: Applying Modal Axiom 5 to ¬g and interpreting 3 as ¬2¬ we obtain
¬2g ⊃2¬2g; standard propositional logic then gives 2g∨2¬2g.

Step 2: By contraposing Anselm’s principle ER1 (i.e., ¬2g ⊃ ¬g) and instantiating
Modal Axiom K with P = ¬2g and Q = ¬g we arrive at 2¬2g ⊃2¬g.

Step 3: The previous steps combine to give 2g∨2¬g which can be rewritten as
2g∨¬3¬¬g and then simplified to 2g∨¬3g.

Conclusion: The previous step combined with God’s possibility ER2 allows us to con-
clude 2g from which Modal Axiom T allows us to infer g.

We can reproduce this level of reasoning in PVS by stating the conclusion of each
step as a lemma as shown below.

PVS text
step1: LEMMA 2 g \/ 2∼2 g

step2: LEMMA 2∼2 g => 2∼g

step3: LEMMA 2 g \/ 2∼g

We then invite PVS to prove each of the lemmas and the conclusion ERC tmp with-
out expanding the modal qualifiers. Here is the beginning of the proof for step1.

PVS proof steps
(lemma "five") (inst - "∼g")

(grind :if-match nil :exclude ("<>" "2") :rewrites "dexpand")

(inst?)

4 Greater automation is possible using modern SMT solvers (Ge and de Moura 2009) but that
can be less helpful when, as here, a proof fails and we want to understand why.

The :exclude construct prevents expansion of the modal qualifiers, but note that
dexpand will automatically rewrite 3 as ¬2¬. This brings us to the following sequent,
which represents the instantiation of Modal Axiom 5 (the formula above the turnstile
line) and allows us to see how this will discharge the conclusion to step1 (the two for-
mulas below the line). That final step is accomplished by the proof command (grind).
Note that the Skolem constant on worlds w!1 comes from the definition of validity for
modal constructs as true in all worlds.

PVS sequent
{-1} NOT (2∼∼val(g))(w!1) IMPLIES (2∼2∼∼val(g))(w!1)

|-------

{1} (2val(g))(w!1)

{2} (2∼2val(g))(w!1)

Preference for one or the other style of mechanically assisted proof depends on the
purpose of the exercise. If it is to examine validity and soundness of a particular argu-
ment (i.e., selection of premises), then I think the automated approach is preferable: it
is generally fast and simple. But if the purpose is to examine a specific proof (i.e., chain
of inferences) then a mechanically assisted reconstruction will be the necessary choice.
I will examine some specific proofs in Section 4 but will favor automated verification
for the remainder of this section.

Returning to the automated verification of Section 3.1.1, PVS assures us that this
specification of the argument is valid, so we should now consider whether the premises
and assumptions are reasonable and sound.

3.1.3 Analysis and Interpretation. We focus first on the assumption of a symmetric
accessibility relation or, alternatively, the standard axiom B.

The argument makes sense only in an Alethic logic, that is one where 2 and 3

are interpreted as necessary and possibly, respectively, and Alethic modal logics are
characterized by having accessibility relations that are equivalence relations. Hence,
the assumption of symmetry in ERC presents no difficulty. Dually, Alethic logics are
also characterized by the standard axioms T, 4, and 5—a combination that is gener-
ally referred to as S55—from which B can be derived as a theorem. We conclude that
the assumption of symmetry or modal axiom B is uncontroversial, provided one ac-
cepts that the Alethic logic needed here is correctly characterized by S5—which few
would dispute. If in doubt, Ladstaetter (2012) provides a clear description of logical,
metaphysical, and physical notions of necessity, and explains why logical necessity is
required here, and why S5 is the right formalization.

We note that the manual proof of Section 3.1.2 used Modal Axioms 5, K, and T
instead of B. However, K is true of all modal logics, and 5 and T are included in S5,
and therefore this combination is acceptable in an Alethic logic.

5 The modal formula K is often mentioned, too, and the logic is then sometimes referred to
as KT45; however, K is a theorem, true in all modal logics (alternatively, it can be seen to
characterize what is a modal logic), so its mention is redundant; furthermore, T and 5 imply 4,
so mention of 4 is also redundant.

Next, we consider the premise ER1. This seems, and is generally considered to be, a
reasonable premise for a truly perfect or greatest being, such as God: if such a being ex-
isted, it would surely not do so merely contingently. Notice that this premise is specific
to the constant g with the interpretation used here: the generalization P ⊃ 2P, where P
is a metavariable ranging over all propositional constants is not a plausible premise, nor
is the instance where g is interpreted as a perfect island, as in Gaunilo’s (circa 1079)
refutation.

The premise ER2 is discussed at length in some philosophical papers (e.g.,
Hartshorne 1961). One concern is that the concept of a perfect or greatest being must be
non self-contradictory (otherwise it could not possibly exist), and this is not straightfor-
ward as there are properties of a perfect being that seem to conflict (e.g., omnibenevo-
lence and perfect justice). Another concern is that non self-contradiction may be insuf-
ficient to ensure the possibility of existence. However, as we are concerned with logic
and mechanization, we accept ER2 as otherwise the proof cannot proceed.

The overall argument seems fairly satisfying: we have one premise about a hypoth-
esized God’s necessary existence, another about his possible existence, and we are able
to conclude his actual existence, which is something of a surprise and therefore satisfy-
ing.

However, we have the background assumption of a symmetric accessibility relation
or the modal axiom B and, in the presence of this assumption, the premise ER1 reduces
to 3g ⊃ g. This is stated in its general form as ER triv in Figure 3; its proof is simply
(grind :polarity? t). But now the argument is exposed as trivial:

3g ⊃ g, 3g, hence g.

We will examine several variants on this argument; mostly they adjust the first
premise, sometimes also the conclusion (using 2g rather than simply g), and some-
times they assume modal axioms other than B. But in all cases, we will see that the
argument reduces to the same trivial form.

Addendum to the published paper (16 August 2020). If M is the modal assumption, P1
and P2 the premises and C the conclusion, then in classical logic the overall deduction
has the form M∧P1∧P2 `C and from this it follows that M ⊃ (P1 ⊃ (P2 ⊃C)). ER triv
is just a modal instance of this inevitable relationship. The reason I label it trivial is
because it is expressed so directly in the premises and conclusion employed.
End of Addendum

3.2 The Version by Kane

Kane (1984) presents a version of the argument that uses a slightly more complex for-
mulation of the first premise. The interpretation of g, the second premise, and the con-
clusion are the same as in Eder and Ramharter’s version from the previous section.

Premise K1: 2(g ⊃2g).
Kane reads this as “Necessarily, if a perfect being exists, then necessarily a perfect
being exists.” He states that Hartshorne and others assume that it follows from

a principle he names N: “By definition, anything which is perfect is such that,
if it exists, it exists necessarily,” where “the first 2 in K1 corresponds to the ‘by
definition’ of N” (Kane 1984, page 337).

Premise K2: 3g (i.e., God’s existence is possible).
Conclusion KC: g (i.e., God exists in a classical sense).

The heart of the PVS specification for this version is shown below.

Kane’s version in PVS
K1: LEMMA 2 (g => 2 g)

K_triv: LEMMA symmetric?(access) IMPLIES 2(P => 2 P) IMPLIES (<> P => P)

K2: AXIOM <> g

KC: THEOREM symmetric?(access) => g

We label the PVS rendition of K1 as a LEMMA rather than an AXIOM because it can be
proved from ER1; the proof is simply (grind :polarity? t). Again, proof of the
conclusion requires modal axiom B or, equivalently, a symmetric accessibility relation.
Given that K1 is entailed by ER1, it is unsurprising that it also reduces to 3g ⊃ g, which
is stated as K triv in the PVS specification. Thus, at bottom, this version is based on
the same trivial reasoning as the previous one.

Kane devotes considerable space to discussing the acceptability of modal axiom B.
But, as explained in the previous section, B is a theorem of S5, which is the standard
formalization for the Alethic logic that is needed to provide the desired interpretations
for 2 and 3, so Kane’s agonizing seems otiose.

3.3 The Version of Malcolm as Reported by Hartshorne

Hartshorne (1961) presents a version of the argument that he attributes to Malcolm
(1960). This has the same premises as Eder and Ramharter’s version, but the conclu-
sion is 2g rather than simply g. Furthermore, rather than modal Axiom B, this version
uses the modal Axiom 5, which is 3P ⊃ 23P. Axiom 5 corresponds to a Euclidean
accessibility relation6 and is assumed in an Alethic logic.

The core of the formal specification in PVS is shown below,

Malcolm’s version in PVS
M1: AXIOM g => 2 g

M_triv: LEMMA Euclidean?(access) IMPLIES (P => 2 P) IMPLIES (<> P => 2 P)

M2: AXIOM <> g

MC: THEOREM Euclidean?(access) => 2 g

MC_alt: THEOREM symmetric?(access) => 2 g

6 A relation R is Euclidean if ∀u,v,w : R(u,v)∧R(u,w) ⊃ R(v,w).

PVS easily proves the conclusion MC, but it also proves the lemma M triv and so again
the reasoning is too trivial to claim our interest.

The conclusion can also be proved assuming modal axiom B or, alternatively, a
symmetric accessibility relation, as stated in MC alt: we first use the method of Eder
and Ramharter’s version to establish g, then a second application of the premise M1
delivers 2g.

3.4 The Version of Adams as Reported by Matthews

Adams (1971) derives a modal argument, not from Proslogion Chapter III, but from
Anselm’s reply (circa 1079, Chapter I, paragraph 6) to Gaunilo (circa 1079). Despite its
different origin, this argument is the same as Kane’s. However, Matthews (2005), who
reports Adam’s version, gives the conclusion as 2g. I will use this latter version as it
completes the set of permutations on the first premise and conclusion.

This version of the argument in PVS is shown below.

Adams’ Version in PVS
A1: LEMMA 2(g => 2 g)

A_triv: LEMMA

Euclidean?(access) IMPLIES 2(P => 2 P) IMPLIES (<> P => 2 P)

A2: AXIOM <> g

AC: THEOREM Euclidean?(access) => 2 g

AC_alt: THEOREM symmetric?(access) => 2 g

As we expect, based on the other variants we have examined, this version can be
proved assuming either a symmetric or Euclidean accessibility relation (i.e., standard
axioms B or 5, respectively). And with a Euclidean accessibility relation, the lemma
A triv reveals that Premise A1 reduces to 3g ⊃ 2g so the conclusion follows trivially
from Premise A2.

3.5 The Version by Hartshorne

Hartshorne (1962, pp. 50–51) presented the first formal reconstruction of the modal ar-
gument. This uses a different and more complex formulation of the first premise than
the others we have examined. The interpretation of g, the second premise, and the con-
clusion are the same as in the versions of Eder and Ramharter, and Kane.

Premise H1: gJ2g, where J denotes strict implication.
Premise H2: 3g (i.e., God’s existence is possible).
Conclusion HC: g (i.e., God exists in a classical sense).

We say that P strictly implies Q if it is not possible for P to be true and Q false: that
is, in an Alethic logic

PJ Q def
= ¬3(P∧¬Q).

It is a theorem that strict implication is the same as necessary material implication:

PJ Q = 2(P ⊃ Q).

This equality is a theorem of all modal logics (i.e., it requires no axioms) but it car-
ries the intended interpretation only in Alethic logics. Thus, premise H1 is equivalent
to Kane’s premise K1 of Section 3.2 and the rest of the argument is also the same as
Kane’s. It follows that the conclusion requires the assumption of a symmetric acces-
sibility relation or, equivalently, modal axiom B, and that under this assumption the
premise H1 reduces to 3g ⊃ g and the argument is thereby seen to be trivial.

In PVS, this is written as follows, where we use |> for strict implication (Rushby
2017).

Hartshorne’s version in PVS
|>(P, Q): pmlformulas = ∼ <>(P & ∼ Q)

H1: AXIOM g |> 2 g

H_triv: LEMMA symmetric?(access) IMPLIES (P |> 2 P) IMPLIES (<> P => P)

H2: AXIOM <> g

HC: THEOREM symmetric?(access) => g

The conclusion HC can be proved by the following PVS commands, whose length is due
to weak quantifier reasoning in PVS.

PVS proof
(lemma "H1") (lemma "H2")

(grind :if-match nil)

(inst?) (skosimp) (inst?)

(grind :polarity? t)

Similarly, H triv is proved as follows.

PVS proof
(grind :if-match nil) (inst? -2) (grind :polarity? t)

We know from Section 3.4 that Adams’ version is the same as Kane’s with the con-
clusion changed to 2g and the modal assumption changed to a Euclidean accessibility
relation or, equivalently, modal axiom 5. Since Hartshorne’s version is equivalent to
Kane’s we can perform the same transformations here to yield the following variant.

Alternative Hartshorne PVS
H_triv_alt: LEMMA

Euclidean?(access) IMPLIES (P |> 2 P) IMPLIES (<> P => 2 P)

HC_alt: THEOREM Euclidean?(access) IMPLIES 2 g

These are proved in the same way as the originals.

Addendum to the published paper (29 July 2021). Harold Thimbleby remarked that
it is a pity that PVS imposes an ASCII rendition and does not reproduce mathematical
notation. I thank Harold for prompting me to report that, in fact, recent versions of
PVS do support extended character sets with mathematical symbols, and all versions of
PVS are able to typeset specifications in LATEX. A LATEX-printed version of Hartshorne’s
version is shown below.

PJ Q: pmlformulas def
= ¬3(P ∧ ¬Q)

H1: AXIOM gJ2g

H1 triv: LEMMA symmetric?(access) ⊃ (PJ2P) ⊃ (3P ⊃ P)

H2: AXIOM 3g

HC: THEOREM symmetric?(access) ⊃ g

Exactly how PVS ASCII text is rendered in LATEX is controlled by a “substitutions”
file pvs-tex.sub. A comprehensive substitution file is provided with PVS but it can
be augmented by the user. The rendition above was generated using the following sub-
stitutions file as augmentation. The first column in this file identifies the ASCII source
to be substituted, the second identifies the kind of PVS object it is (see the PVS docu-
mentation), the third gives the size of the substitution in ems, and the final column gives
the desired LATEX substitution.

PVS LATEX substitution file
|> 2 3 {#1 \strictif #2}

|> id 1 \strictif

H1 id 2 \pvsid{H1}

H1_triv id 3 \pvsid{H1\_triv}

H2 id 2 \pvsid{H2}

IMPLIES id 3 \supset

= key 2 \defn

˜ id 1 \sim

˜ id 1 \neg

=> id 1 \supset

& id 1 \wedge

It might be considered that these substitutions are too aggressive because they fail
to distinguish those connectives that are to be interpreted modally from those that are
propositional. If we remove the last three lines from the substitution file above (where
the third-last line was overriding the fourth-last), we restore this distinction and generate
the following rendition.

PJ Q: pmlformulas def
= ∼3(P & ∼Q)

H1: AXIOM gJ2g

H1 triv: LEMMA symmetric?(access) ⊃ (PJ2P) ⊃ (3P ⇒ P)

H2: AXIOM 3g

HC: THEOREM symmetric?(access) ⇒ g

End of Addendum

4 The Utility of These Methods
In my view, the utility of the methods illustrated here, and of computational philosophy
more generally, is not so much that they enable verification of hard logical problems
(those abound in logic and mathematics, not so much in philosophy) but that they en-
able rapid and error-free exploration of multiple formulations of the same or similar
problems. This allows discovery of equivalences and subtle differences, not to mention
checking of validity and discovery of economical and persuasive formulations.

In the case of the five specific formulations of the Modal Ontological Argument ex-
amined in Section 3, we found that all of them are variations on a simple pattern. Their
authors and subsequent commentators did not seem to recognize this similarity, nor
some of the differences within it. Hartshorne’s (1962) formulation uses strict implica-
tion whereas his earlier (1961) treatment of Malcolm’s (1960) original exposition does
not. Eder and Ramharter (2015, Section 4.1) state that they follow Hartshorne’s (1962)
formulation, but then use material rather than strict implication. Kane (1984) states that
his formulation is from Hartshorne (1962) but “readers of that work and subsequent
journal literature on the modal OA may not recognize the adaptation.” He attributes
it to an unpublished simplification due to C. Anthony Anderson (Kane 1984, p. 339).
Adams (1971) treatment is derived not from the Proslogion, but from Anselm’s reply
to Gaunilo.

Despite the apparent differences in their original presentations, all of the reconstruc-
tions have a first premise that asserts, in one way or another, that if a perfect or greatest
being (i.e., God) exists, then he exists necessarily (i.e., g ⊃2g, 2(g ⊃2g), or gJ2g);
a second premise that asserts his existence is possible (i.e., 3g); and a conclusion that
he exists or necessarily exists (i.e., g or 2g).

Proof that the conclusion follows from the premises requires certain properties of
the modal logic concerned; these can be expressed either by citing one of the standard
modal axioms (i.e., B or 5) or the corresponding property of the accessibility relation
on possible worlds (i.e., symmetric or Euclidean, respectively). The argument requires
an Alethic logic (one where the modalities express necessity and possibility) and these
have accessibility relations that are equivalence relations or, alternatively, assume modal
axioms T and 5, and have B as a theorem; consequently, there can be no objection to
the modal properties required in the proofs.

In addition to verifying the different formulations of the argument, we showed that
in each case, the modal axiom that was assumed is sufficient to prove that the first

premise entails a formula that states that the second premise directly entails the conclu-
sion, that is:

Modal axiom ⊃ (Premise 1 ⊃ (Premise 2 ⊃ Conclusion)).

Thus, the reasoning is essentially trivial.
Mechanization is needed in this kind of examination for the same reasons that auto-

mated calculation is used in other branches of engineering and science: it is simply too
tedious and too error-prone to conduct repetitive analyses by hand. Some may concede
this for hard problems but dismiss it for the Modal Ontological Argument: one reviewer
wrote “the results of the analysis have been common knowledge in the field for at least
the past forty years (at least among those with even minimal competence in proposi-
tional modal logic).” I would respectfully disagree that these results have been common
knowledge and would also be cautious about widespread reliability, if not competence,
in modal logic. To quote Lewis (1970, p. 175):

“Philosophy abounds in troublesome modal arguments—endlessly debated,
perennially plausible, perennially suspect. The standards of validity for modal
reasoning have long been unclear; they become clear only when we provide a
semantic analysis of modal logic by reference to possible worlds. . . ”

For illustration, this journal published a paper by Jacquette (1997) in which he crit-
icizes Hartshorne’s (1962) formulation of the Modal Ontological Argument. Jacquette
presents a semi-formal 10-line proof that he attributes to Hartshorne and finds it defec-
tive on several grounds. Most damagingly, he claims there is a

“logical difficulty in a key assumption of the proof that renders the entire in-
ference unsound. Proposition (5), which Hartshorne says follows from (1) as a
modal form of modus tollens, is clearly false.”

Here, Proposition (1) is Anselm’s Principle g ⊃ 2g and (5) is 2¬2g ⊃ 2¬g which,
contrary to Jacquette, is a perfectly valid deduction from (1), as can be verified by
adding it to any of the PVS analyses presented in Section 3.

Jacquette attributes what he considers Hartshorne’s error to his use of a “modal form
of modus tollens” that Jacquette writes as (α→ β)→ (2 ∼ β → 2 ∼ α) and states “is
not generally true.” He presents a counterexample in which α is “snow is red” and β is
“2+2 = 5.” This single formula highlights a couple of interesting pitfalls in semi-formal
use of modal logic. First of all, Hartshorne uses strict implication, which we write as J
to distinguish it from the ⊃ of material implication. Jacquette makes no mention of this
and it is not clear whether his→ is intended to represent strict or material implication.

Second, it is often necessary to be careful about which parts of a sentence are to
be interpreted modally, and which are conventional propositional logic. Specifically,
a correct statement of modal modus tollens is expressed in PVS as follows (i.e., two
modal sentences connected propositionally),

PVS text
tollens: LEMMA (P => Q) IMPLIES (2∼Q => 2∼P)

whereas the following (i.e., a single modal sentence) is invalid.

PVS text
tollens_bad: CLAIM (P => Q) => (2∼Q => 2∼P)

Jacquette seems to be using the invalid form, but we cannot be sure because his notation
lacks the necessary distinctions. The key difference is that the correct form quantifies
possible worlds separately for each side of the implication, whereas the invalid form
requires the whole sentence to be true in the same worlds. This is similar to the reason
that the deduction theorem is generally invalid in modal logic.

These errors in Jacquette’s analysis are easily detected using truly formal and me-
chanically checked specifications. Furthermore, we or Jacquette could easily satisfy
ourselves that even if we have qualms about Hartshorne’s proof (i.e., selection of infer-
ence steps), his formulation of the argument (i.e., selection of premises and conclusion)
is valid.

Although Jacquette’s charge of invalidity for Hartshorne’s formulation of the argu-
ment is mistaken, he raises other points that have merit. In particular, he observes that
Anselm’s presentation has the form of a reductio ad absurdum whereas Hartshorne’s
(and we might add, all others we have considered) does not. He also notes that Anselm
uses the notion of “greater than” (i.e., an ordering) on beings and that he uses “con-
ceivable” as a modality that seems different than the “possible” of alethic logic. These
points are all abstracted away in conventional formalizations of the argument. Accord-
ingly, Jacquette (1997, Section 4) uses ideas that he attributes to Priest (2002) to con-
struct a formalization of the argument that remedies these deficiencies. His treatment
begins as follows.

1. [Definition of g:] g = δx : ¬∃y : 3(y > x).
Here, δ is “indifferently a definite or indefinite descriptor” (i.e., “the” or “an”) and I
have substituted 3 for Jacquette’s τ, which he intends as “an intensional modality”
(since no semantics are given, this shift to a more familiar notation has no impact).

An informal reading is that God is some being than which no greater is conceiv-
able. The construction ¬∃y : 3(y > x) is also used in formalizations using first order
modal logic for the traditional, Proslogion Chapter II, Ontological Argument (Eder and
Ramharter 2015, Section 4.2). These modal formalizations of the traditional argument
have been mechanized in PVS (Rushby 2019, Section 3.1), as has a version in clas-
sical first-order logic that uses definite descriptions (Rushby 2013), where indefinite
descriptions or choice functions, and Hilbert’s ε, are also discussed.

We might wonder whether the given construction is equivalent to ¬3∃y : (y > x)
(i.e., interchange the modality and quantifier) and the answer is: it depends on whether
we have constant or varying domains and, if varying, whether they are nonincreasing or
nondecreasing, thereby highlighting some of the subtlety lurking here (Rushby 2017,
Section 3).

The next interesting line in Jacquette’s treatment is the following.

3. [Conceivable greatness:] ∀x : (¬E!x ⊃ ∃y : 3(y > x)).
Here, E!x is a predicate indicating the “real existence” of x.

Jacquette uses the definition 1 and premise 3 in a reductio proof that establishes E!g:
that is, the real existence of God.

Now, we could attempt to verify his proof in PVS (in fact, I have done this) but I am
reluctant, this late in the present paper, to extend description of the PVS embedding of
modal logic from propositional to first order, so I will make my point using Jacquette’s
informal notation, but the following derives from experience applying first order modal
reasoning in PVS to formal verification of the Proslogion Chapter II argument, where
the methods are fully described (Rushby 2019).

First, Jacquette does not indicate whether g and E!x are to be interpreted flexibly
(i.e., dependent on the world) or rigidly (i.e., independent of it). I assume a rigid inter-
pretation. Then, nowhere in Jacquette’s proof is the construction ∃y : 3(y > x) opened
up to reveal either y or the > relation. Thus, we can replace the whole thing by an unin-
terpreted function F(x). Then 1 and 3 above become the following “primed” variants;

1′: g = δx : ¬F(x),
3′: ∀x : ¬E!x ⊃ F(x),

and we can replace 3′ by its contrapositive

3′′: ∀x : ¬F(x) ⊃ E!x.

But now the deduction is revealed as trivial: we have ¬F(g) from 1′ and instantiate 3′′ to
give ¬F(g) ⊃ E!g and the conclusion E!g is immediate. Furthermore, the proof has lost
all trace of modal reasoning. Jacquette’s goal was to make explicit more of Anselm’s
reasoning, but he ends up trivializing it.

Although his execution is defective, we can agree with Jacquette’s motivation that a
faithful reconstruction of Anselm’s argument should represent his language explicitly.
A reviewer for this paper makes the same point and suggests that PVS could be used
systematically to explore different renditions for “conceive,” “nothing greater” etc. I
am sympathetic to this but suspect that the defective renditions would exhibit triviality
in the style of Jacquette’s example rather than invalidity, and this is more difficult to
detect automatically. I hope others may wish to explore some of the alternatives and
that verification tools such as PVS will assist their endeavors.

5 Conclusion
We have examined several reconstructions of the modal ontological argument from
Anselm’s Proslogion Chapter III and his reply to Gaunilo. When fully formalized, all
of the reconstructions have a very similar form, and are proved in the same way.

The proofs require certain properties of the modal logic employed, and these can
be specified either by citing one of the standard modal axioms (i.e., B or 5), or the
corresponding property of the accessibility relation on possible worlds (i.e., symmetric
or Euclidean, respectively). The argument requires an Alethic logic (one where the
modalities express necessity and possibility) and these have accessibility relations that
are equivalence relations or, alternatively, assume modal axioms T and 5, and have B as
a theorem; consequently, there can be no objection to the modal properties required in
the proofs.

We expressed the various reconstructions in PVS and mechanically verified their
validity. We also showed that in each case, the modal axiom that was assumed is suffi-
cient to prove that the first premise entails a formula that states that the second premise

directly entails the conclusion, that is:

Modal axiom ⊃ (Premise 1 ⊃ (Premise 2 ⊃ Conclusion)).7

Thus, the reasoning is essentially trivial and standard readings of the modal ontological
argument should not retain our interest.

We also showed how mechanically assisted formalization and analysis could expose
flawed treatments, including one published in this journal, and also vacuous formaliza-
tions that present the appearance of Anselm’s argument but are logically trivial.

Campbell (2018, 2019) provides an alternative reading in which Chapter III of the
Proslogion does not stand alone but continues the argument of Chapter II. Campbell
interprets Anselm differently than the scholars considered here and constructs a differ-
ent translation into English. Formal analysis of his and related readings will require
separate treatment.

This paper is the fifth is a series where I apply mechanized verification to analy-
sis of formal reconstructions of Anselm’s Ontological arguments. In the first (Rushby
2013), I examined a formalization by Oppenheimer and Zalta (2011) for the traditional
Proslogion II argument that uses definite descriptions. I showed that this formalization
is very close to circular (i.e., begs the question) in an informal sense. In the second
(Rushby 2018), I gave formal definitions for question begging in formal proofs and I
showed that all the examined reconstructions of the Proslogion II argument in classical
(i.e., first- and higher-order) logic were guilty of begging the question. In a revision and
extension to that paper (Rushby 2020), I additionally showed that all these reconstruc-
tions entail variants that apply no interpretation to “something than which there is no
greater” and are therefore vacuous and vulnerable to Gaunilo’s (circa 1079) refutation,
and I argued that the basic reconstructions inherit these charges. In the fourth (Rushby
2019), I extended these analyses to reconstructions of the Proslogion II argument in
quantified modal logic and found they suffer the same defects. In the present paper, we
have examined reconstructions of the Proslogion III argument in propositional modal
logic and found it to be trivial once the assumed modal axiom is taken into account.

Mechanization is necessary for these analyses because they require detailed scrutiny
of small variations in premises and assumptions. This is tedious and error-prone to do
by hand, but fast, simple, and even enjoyable to accomplish with mechanized assistance.
I hope these exercises may encourage others to apply modern formal verification tools
to examination of other suitable philosophical and theological topics.

for discussions on modal logic and its automation. This work was funded by SRI Inter-
national and by my retirement plan.

7 This relationship is inevitable in a deduction with only two premises. What is surprising, and
the reason I label it trivial, is how directly it is expressed by the premises employed.

